x^5-5 x + 12 の厳密解 with *Mathematica* 13.3

part2. Galois群がD_5 の方程式

2025 年2月 by mixedmoss

§0. [準備] 原始元，分解方程式，ガロア群

Part1 からの輸入です

In[2208]:=

```
ClearAll["`*"]
f[x_] = x^5 - 5 x + 12;
```

f(x)=0の解を x1, x2, x3, x4, x5 . f(x)=0 の原始元の一つをv = 5 x1+2 x2 + 3 x3 + 4 x4 + 1 x5 (part2 では, part1 と異なり α,β... などの代わりに x1,x2,...を利用する).　vの最小多項式をV[x] とすると

In[2210]:=

$$V[x\_] = 148\,996\,000 - 1\,521\,500\, x^2 - 12\,500\, x^4 + 3325\, x^6 - 110\, x^8 + x^{10};$$

また x1, x2, x3, x4, x5 を原始元vで表すと

In[2211]:=

$$\{x1, x2, x3, x4, x5\} = \Big\{\frac{1}{9\,239\,628\,268\,800}\big(5\,117\,531\,026\,400 + 2\,122\,210\,541\,600\, v - 92\,467\,168\,500\, v^2 - 16\,386\,746\,700\, v^3 - 6\,336\,236\,320\, v^4 + 421\,422\,680\, v^5 + 183\,059\,535\, v^6 - 991\,635\, v^7 - 1\,592\,443\, v^8 - 78\,265\, v^9\big),$$
$$\frac{1}{923\,962\,826\,880}\big(-201\,405\,134\,560 - 37\,539\,305\,120\, v - 7\,761\,390\,780\, v^2 - 3\,277\,349\,340\, v^3 + 993\,991\,688\, v^4 + 84\,284\,536\, v^5 - 12\,606\,567\, v^6 - 198\,327\, v^7 - 7141\, v^8 - 15\,653\, v^9\big), \frac{-16\,080\,205\,600 + 881\,249\,100\, v^2 - 18\,671\,920\, v^4 - 295\,305\, v^6 + 8621\, v^8}{23\,936\,860\,800},$$
$$\frac{1}{923\,962\,826\,880}\big(-201\,405\,134\,560 + 37\,539\,305\,120\, v - 7\,761\,390\,780\, v^2 + 3\,277\,349\,340\, v^3 + 993\,991\,688\, v^4 - 84\,284\,536\, v^5 - 12\,606\,567\, v^6 + 198\,327\, v^7 - 7141\, v^8 + 15\,653\, v^9\big),$$
$$\frac{1}{9\,239\,628\,268\,800}\big(5\,117\,531\,026\,400 - 2\,122\,210\,541\,600\, v - 92\,467\,168\,500\, v^2 + 16\,386\,746\,700\, v^3 - 6\,336\,236\,320\, v^4 - 421\,422\,680\, v^5 + 183\,059\,535\, v^6 + 991\,635\, v^7 - 1\,592\,443\, v^8 + 78\,265\, v^9\big)\Big\};$$

f(x) のGalois群を Gal , σ=(1,2,3,4,5), τ=(2,3,5,4) とすると, τ^2=(2,5)(3,4)

$\mathrm{Gal} = \{Id, \sigma, \sigma^2, \sigma^3, \sigma^4, \tau^2\sigma, \tau^2\sigma^2, \tau^2\sigma^3, \tau^2\sigma^4\} = D_5 = C_5 + \tau^2 C_5$

ただし，C_5 は5次巡回群，D_5 は10次の2面体群で，分解列は　$\mathrm{Gal} = D_5 \triangleright C_5 \triangleright \{Id\}$

§1．Lagrange分解式 s1,s2,s3,s4 の5乗の値を求める

$C_5 \triangleright${Id}の部分に注目し，ラグランジュの5次分解式をつくる .(ζは1の原始 5 乗根のひとつ.)
このとき，V(v)と円分方程式 $v^4+v^3+v^2+v+1$ で割った余りを考える.（PolynomialMod)

In[2212]:=

```
s0 = x1 + x2 +  x3 + x4 + x5 // PolynomialMod[#, V[v]] &;
s1 =
  x1 + ξ x2 + ξ^2 x3 + ξ^3 x4 + ξ^4 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
s2 = x1 + ξ^2 x2 + ξ^4 x3 + ξ x4 + ξ^3 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
s3 = x1 + ξ^3 x2 + ξ x3 + ξ^4 x4 + ξ^2 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
s4 = x1 + ξ^4 x2 + ξ^3 x3 + ξ^2 x4 + ξ x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
```

解と係数の関係より s0 = 0. また

$$(\#)\begin{cases} \text{s1を}\sigma\text{で移すと s1} \to \zeta^4\,\text{s1}, & \tau^2\ \text{で移すと s1} \to \text{s4} \\ \text{s2を}\sigma\text{で移すと s2} \to \zeta^3\,\text{s2}, & \tau^2\ \text{で移すと s2} \to \text{s3} \\ \text{s3を}\sigma\text{で移すと s3} \to \zeta^2\,\text{s3}, & \tau^2\ \text{で移すと s3} \to \text{s2} \\ \text{s4を}\sigma\text{で移すと s4} \to \zeta\,\text{s4}, & \tau^2\ \text{で移すと s4} \to \text{s1} \end{cases}$$

即ち, τ^2 は 分解式を「s1↔s4, s2↔s3」のように移し, σは「s1 → ξ^4 s1→ → ξ^3 s1→ → ξ^2 s1 → ξ s1 → s1」のように移す.
さらに 同型写像 μ: $\zeta \to \zeta^3$ とすると, 分解式は次のように移る.

(by μ) s1→s3→s4→s2 →s1⋯ (by μ^2) s1↔s4, s2↔s3

さて(#)より $s1^5$, $s2^5$,$s3^5$, $s4^5$は σで不変. t1=$s1^5+s4^5$, t2=$s2^5+s3^5$, t3=$s1^5\,s4^5$, t4 = $s2^5\,s3^5$ はτ^2 で不変
故に t1,t2,t3,t4 はD_5 で不変なので, D_5の固定体 Q(ζ)に入るはず. 「単純計算」で

In[2217]:=

```
t1 = s1^5 + s4^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
t2 = s2^5 + s3^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
t3 = s1^5 * s4^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
t4 = s2^5 * s3^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
```

Out[2217]=

$-8750 - 5000\,\xi^2 - 5000\,\xi^3$

Out[2218]=

$-3750 + 5000\,\xi^2 + 5000\,\xi^3$

Out[2219]=

$78125 + 156250\,\xi^2 + 156250\,\xi^3$

Out[2220]=

$-78125 - 156250\,\xi^2 - 156250\,\xi^3$

【参考】μ^3: $\zeta \to \zeta^2$ で t1 を移すと, t2=-8750-5000($\xi^4+\xi$)=-3750+5000 ξ^2 + 5000 ξ^3
同様にして t3 の式から 計算なしに t4 の式が得られる.

In[∘]:=

ここで ξの偏角を（±2 π / 5）に取っても一般性を失わない.
このとき「$\xi^2+\xi^3 = 1/2\left(-1-\sqrt{5}\right)$」だから
$s1^5 + s4^5 = -6250 + 2500\sqrt{5}$, $s2^5 + s3^5 = -6250 - 2500\sqrt{5}$,
$s1^5\,s4^5 = -78125\sqrt{5}$, $s2^5\,s3^5 = 78125\sqrt{5}$ -- (1)

p=$s1^5$,q=$s2^5$,r=$s3^5$,s=$s4^5$とおくと, (1)より

p+s=$-6250+2500\sqrt{5}$, ps=$-78125\sqrt{5}$, q+r=$-6250-2500\sqrt{5}$, qr=$78125\sqrt{5}$

「p, s を 2 解とする 2 次方程式」と「q, r を 2 解とする 2 次方程式」を作ることができるので p,q,r,sの値が求まる．　理論的に解けることが分かったのでSolveを使って{p,q,r,s}を求め 変数[pqrs]に入れる．

In[2221]:=

```
pqrs = {p, q, r, s} /. Solve[{p + s == -6250 + 2500 Sqrt[5], p s == -78125 Sqrt[5],
    q + r == -6250 - 2500 Sqrt[5], q r == 78125 Sqrt[5]}, {p, q, r, s}] // FullSimplify
```

Out[2221]=

$$\Big\{\Big\{-125\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right),\ 125\left(-25-10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),$$
$$625\left(-5-2\sqrt{5}-3\sqrt{5+\frac{11}{\sqrt{5}}}\right),\ 125\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)\Big\},$$
$$\Big\{125\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right),\ 125\left(-25-10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),$$
$$625\left(-5-2\sqrt{5}-3\sqrt{5+\frac{11}{\sqrt{5}}}\right),\ -125\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)\Big\},$$
$$\Big\{-125\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right),\ -125\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),$$
$$125\left(-25-10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),\ 125\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)\Big\},$$
$$\Big\{125\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right),\ -125\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),$$
$$125\left(-25-10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right),\ -125\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)\Big\}\Big\}$$

ここで

In[2222]:=

$$625\left(-5-2\sqrt{5}-3\sqrt{5+\frac{11}{\sqrt{5}}}\right) == -125\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)$$

Out[2222]=

```
True
```

であるから，(p,q,r,s)の組は4 組あるが，これらは順序が異なっているだけで，最初の組を(p,q,r,s)と改めておくと，全ての解は (p,q,r,s),(s,q,r,p),(p,r,q,s),(s,r,q,p)の4通り．(実は x1,x2,x3,x4の値は (p,q,r,s)を(s,r,q,p),(r, p,s,q), (q,s,p,r)に変えても変わらないので，これらの解は実質 2 組.)

§2.(s1 s2^2+s3^2 s4) と (s1^2 s3+s2 s4^2) の値を考慮し s1,s2,s3,s4 の値を求める

ここで s0,s1,s2,s3,s4の定義とs0=0から

$$(\#\#)\begin{cases} x1 &= 1/5\,(s1 + s2 + s3 + s4) \\ x2 &= 1/5\left(\zeta^4\, s1 + \zeta^3\, s2 + \zeta^2\, s3 + \zeta\, s4\right) \\ x3 &= 1/5\left(\zeta^3\, s1 + \zeta\, s2 + \zeta^4\, s3 + \zeta^2\, s4\right) \\ x4 &= 1/5\left(\zeta^2\, s1 + \zeta^4\, s2 + \zeta\, s3 + \zeta^3\, s4\right) \\ x5 &= 1/5\left(\zeta\, s1 + \zeta^2\, s2 + \zeta^3\, s3 + \zeta^4\, s4\right) \end{cases}$$

w1=s1 s2^2+s2 s4^2 と w2 = s1^2 s3+s2 s4^2 は σとτ で変わらないので，D_5 の不変体Q(ζ)に入る．「単純計算」して

In[2223]:=

```
s1 s2^2 + s3^2 s4 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
s1^2 s3 + s2 s4^2 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
```

Out[2223]=

$-50 - 100\,\xi^2 - 100\,\xi^3$

Out[2224]=

$50 + 100\,\xi^2 + 100\,\xi^3$

【参考】μ^3 : $\xi \to \xi^2$ で w1の式 を移すと

$s2\,s4^2 + s1^2\,s3 = -50 - 100\,\xi^4 - 100\,\xi = -50 - 100\left(-1 - \xi^2 - \xi^3\right) = 50 + 100\,\xi^2 + 100\,\xi^3$

このように計算無しで w1の式からw2 の式が得られる.

「$\xi^2 + \xi^3 = 1/2\left(-1 - \sqrt{5}\right)$」だから,

$\left(s1\,s2^2 + s3^2\,s4,\ s1^2\,s3 + s2\,s4^2\right) = \left(50\sqrt{5},\ -50\sqrt{5}\right)$ -- (2)

変数[pqrs]には $\left(s1^5,\ s2^5,\ s3^5,\ s4^5\right)$ の値が4組入っている.ここで s1, s2, s3, s4は実数としても一般性を失わない（補足参照).

このとき（2）を満たす（s1, s2, s3, s4）は次の2組となる.

In[2225]:=

```
realRoot[p_] := Sign[p] Abs[p]^(1/5) (*実数に対し実5乗根を求めるコマンド*)
realRoot /@ pqrs;
s1s2s3s4 = Select[%, (FullSimplify[ #[[1]] * #[[2]]^2 + #[[3]]^2 * #[[4]]] == 50 Sqrt[5] &&
     FullSimplify[#[[1]]^2 * #[[3]] + #[[2]] * #[[4]]^2] == -50 Sqrt[5]) &]
```

Out[2227]=

$$\Bigg\{\Bigg\{-5^{3/5}\left(25 - 10\sqrt{5} + 3\sqrt{125 - 55\sqrt{5}}\right)^{1/5},\ -5^{3/5}\left(25 + 10\sqrt{5} - 3\sqrt{125 + 55\sqrt{5}}\right)^{1/5},$$
$$-5^{4/5}\left(5 + 2\sqrt{5} + 3\sqrt{5 + \frac{11}{\sqrt{5}}}\right)^{1/5},\ 5^{3/5}\left(-25 + 10\sqrt{5} + 3\sqrt{125 - 55\sqrt{5}}\right)^{1/5}\Bigg\},$$
$$\Bigg\{5^{3/5}\left(-25 + 10\sqrt{5} + 3\sqrt{125 - 55\sqrt{5}}\right)^{1/5},\ -5^{3/5}\left(25 + 10\sqrt{5} + 3\sqrt{125 + 55\sqrt{5}}\right)^{1/5},$$
$$-5^{3/5}\left(25 + 10\sqrt{5} - 3\sqrt{125 + 55\sqrt{5}}\right)^{1/5},\ -5^{3/5}\left(25 - 10\sqrt{5} + 3\sqrt{125 - 55\sqrt{5}}\right)^{1/5}\Bigg\}\Bigg\}$$

§3.　x1,x2,x3,x4,x5 の値を求める

これらの2組の最初の解を {s1, s2,s3,s4}とおくと，他の解は {s4,s3,s2,s1}．ところが，この2組の置換では {x2,x3,x4,x5}の順序が変わるだけなので，同じ解集合を与える．故に，解集合{x1,x2,x3,x4,x5} の値は，次の {x1’,x2’,x3’,x4’,x5’}となる．

この計算はバグることが多いです．結果がおかしいときは「評価を放棄」して「このセルのみ再評価」して下さい．

In[2240]:=

```
{s1', s2', s3', s4'} = s1s2s3s4[[1]];
ξ = 1/4 (-1 + √5 + √(2 (5 + √5)) I);(*=Cos[2π/5]+i Sin[2π/5],最後に代入するのが吉*)

x1' = 1 / 5 (s1' + s2' + s3' + s4') // Simplify
x2' = 1 / 5 (ξ^4 s1' + ξ^3 s2' + ξ ^2 s3' + ξ s4') // Simplify
x3' = 1 / 5 (ξ^3 s1' + ξ s2' + ξ^4 s3' + ξ^2 s4') // Simplify
x4' = 1 / 5 (ξ^2 s1' + ξ^4 s2' + ξ s3' + ξ^3 s4') // Simplify
x5' = 1 / 5 (ξ s1' + ξ^2 s2' + ξ ^3 s3' + ξ^4 s4') // Simplify
```

Out[2242]=

$$-\frac{1}{5^{2/5}}\left(\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}-\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}+\left(25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}\right)^{1/5}+\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)^{1/5}\right)$$

Out[2243]=

$$\frac{1}{256\times 5^{2/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(64\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}-\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3}-4\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2}\left(25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}\right)^{1/5}-16\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)^{1/5}\right)$$

Out[2244]=

$$\frac{1}{256\times 5^{2/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(16\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)-4\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2}-64\left(25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}\right)^{1/5}-\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3}\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)^{1/5}\right)$$

Out[2245]=

$$\frac{1}{256\times 5^{2/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(-16\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)+4\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2}-\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3}\left(25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}\right)^{1/5}-64\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)^{1/5}\right)$$

Out[2246]=

$$\frac{1}{256\times 5^{2/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(-64\left(25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}+\left(-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}\right)^{1/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3}-16\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}\right)^{1/5}-4\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2}\left(25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}\right)^{1/5}\right)$$

近似値を使い f (x) = 0 の解と比較する．まず 上で求めた解の近似値は

In[2235]:=

```
{Dynamic[N[x1']], Dynamic[N[x2']], Dynamic[N[x3']], Dynamic[N[x4']], Dynamic[N[x5']]}
```

Out[2235]=

```
{-1.84209, 1.2729 + 0.719799 i,
 -0.351854 + 1.70956 i, -0.351854 - 1.70956 i, 1.2729 - 0.719799 i}
```

一方, f (x) = 0 の解の近似値を求めると

In[2236]:=

```
x /. NSolve[f[x] == 0, x]
```

Out[2236]=

```
{-1.84209, -0.351854 - 1.70956 i,
 -0.351854 + 1.70956 i, 1.2729 - 0.719799 i, 1.2729 + 0.719799 i}
```

少なくとも小数点以下5桁までは一致する．また少し時間はかかるが f(x1)=0となることが確かめられる．(興味のある方は下のコメントを外して実行してみてください)

In[2237]:=

```
(*f[FullSimplify[x1']]//FullSimplify*)
```

虚数根の方は「とても」時間がかかるので省略する．なお実数根x1と虚数根x2を 累乗根を使って表すと,

In[2238]:=

```
Dynamic[x1'] // TraditionalForm // Print
```

$$-\frac{1}{5^{2/5}}\left(\sqrt[5]{25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}}-\sqrt[5]{-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}}+\sqrt[5]{25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}}+\sqrt[5]{25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}}\right)$$

In[2239]:=

```
Dynamic[x2'] // TraditionalForm // Print
```

$$\frac{1}{256\times 5^{2/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\left(64\sqrt[5]{-25+10\sqrt{5}+3\sqrt{125-55\sqrt{5}}}-\sqrt[5]{25-10\sqrt{5}+3\sqrt{125-55\sqrt{5}}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3}-4\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2}\sqrt[5]{25+10\sqrt{5}-3\sqrt{125+55\sqrt{5}}}-16\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)\sqrt[5]{25+10\sqrt{5}+3\sqrt{125+55\sqrt{5}}}\right)$$

§4. まとめ

Gal = D_5 の有理数係数 5 次方程式(4次の係数は0)に対する解法のまとめです.

[Step1] 5次のLagrangeの分解式 s_0,s_1,s_2,s_3,s_4 を作る (ζは 1 の原始 5 乗根の一つ)

$$\begin{cases} s_0 &= x_1+x_2+x_3+x_4+x_5\ (=0) \\ s_1 &= x_1+\zeta x_2+\zeta^2 x_3+\zeta^3 x_4+\zeta^4 x_5 \\ s_2 &= x_1+\zeta^2 x_2+\zeta^4 x_3+\zeta x_4+\zeta^3 x_5 \\ s_3 &= x_1+\zeta^3 x_2+\zeta x_3+\zeta^4 x_4+\zeta^2 x_5 \\ s_4 &= x_1+\zeta^4 x_2+\zeta^3 x_3+\zeta^2 x_4+\zeta x_5 \end{cases} \Rightarrow \begin{cases} x_1 &= s_1+s_2+s_3+s_4 \\ x_2 &= \zeta^4 s_1+\zeta^3 s_2+\zeta^2 s_3+\zeta s_4 \\ x_3 &= \zeta^3 s_1+\zeta s_2+\zeta^4 s_3+\zeta^2 s_4 \\ x_4 &= \zeta^2 s_1+\zeta^4 s_2+\zeta s_3+\zeta^3 s_4 \\ x_5 &= \zeta s_1+\zeta^2 s_2+\zeta^3 s_3+\zeta^4 s_4 \end{cases}$$

ここで 次の関係が成り立つ.

$$(\#)\begin{cases} s_1\text{ を}\sigma\text{で移すと } s_1 \to \zeta^4 s_1, & \tau^2\text{ で移すと } s_1 \to s_4 \\ s_2\text{ を}\sigma\text{で移すと } s_2 \to \zeta^3 s_2, & \tau^2\text{ で移すと } s_2 \to s_3 \\ s_3\text{ を}\sigma\text{で移すと } s_3 \to \zeta^2 s_3, & \tau^2\text{ で移すと } s_3 \to s_2 \\ s_4\text{ を}\sigma\text{で移すと } s_4 \to \zeta s_4, & \tau^2\text{ で移すと } s_4 \to s_1 \end{cases}$$

τ^2 は 分解式を「$s_1 \leftrightarrow s_4$, $s_2 \leftrightarrow s_3$」のように移し, σは「$s_1 \to \zeta^4 s_1 \to \to \zeta^3 s_1 \to \to \zeta^2 s_1 \to \zeta s_1 \to s_1$」のように移す.

さらに 同型写像 μ: $\zeta \to \zeta^3$ とすると, 分解式は次のように移る.

(##) (by μ) $s_1 \to s_3 \to s_4 \to s_2 \to s_1$ (by μ^2) $s_1 \leftrightarrow s_4$, $s_2 \leftrightarrow s_3$

$t_1 = s_1^5 + s_4^5$, $t_2 = s_2^5 + s_3^5$, $t_3 = s_1 s_4$, $t_4 = s_2 s_3$ とおく.
t_1, t_2, t_3, t_4 は D_5 で不変なので, D_5の固定体 $Q(\zeta)$ に入り t_1 ,t_2,t_3,t_4 の値は「単純計算」で求まる. 2 次方程式を解くことにより $(s_1^5, s_2^5, s_3^5, s_4^5)$ の値が 4 組求まる.
ここで「$s_1^5, s_2^5, s_3^5, s_4^5$ は全て実数」または「s_1と s_4, s_2と s_3が複素共役」（補足参照)

[Step2] $(s_1 s_2^2 + s_3^2 s_4)$ と $(s_1^2 s_3 + s_2 s_4^2)$ の値を考慮し s_1, s_2, s_3, s_4 の値を求め, さらに x_1, x_2, x_3, x_4, x_5 を求める.

$u_1 = s_1 s_4$, $u_2 = s_2 s_3$, $w_1 = s_1 s_2^2 + s_3^2 s_4$,
$w_2 = s_1^2 s_3 + s_2 s_4^2$ とおくと, (♯) より u_1, u_2, w_1, w_2 は, D_5 の不変体 Q (ζ) に入る.
故に「単純計算」で u_1, u_2, w_1 と w_2 の値は求まる.

(ア) $s_1^5, s_2^5, s_3^5, s_4^5$ が実数の時, $s_1, s_2, s_3,$
s_4 が全て実数としても一般性を失わない.(補足3). 故に w_1, w_2 を満足させる (s_1, s_2, s_3, s_4) はすぐ求まる.

(イ) s_1 と s_4 , s_2 とs_3 が複素共役のとき, C_5 のときと同様に極形式を使えば, u_1, u_2 の値から絶対値が, w_1 とw_2 の値から偏角が求まる. これにより (s_1, s_2, s_3, s_4) の値が求まる.

---

§5. 補足

[補足1] $s_1 s_4$, $s_2 s_3$, $s_1 s_2^2 + s_3^2 s_4$, $s_1^2 s_3 + s_2 s_4^2$, $s_1^5 + s_4^5$, $s_2^5 + s_3^5$ は Q $(\sqrt{5})$ に入る

写像μ^2 : $\zeta \to \zeta^4$ により「$s_1 \leftrightarrow s_4$, $s_2 \leftrightarrow s_3$」. 故に $s_1 s_4$, $s_2 s_3$,
$s_1 s_2^2 + s_3^2 s_4$, $s_1^2 s_3 + s_2 s_4^2$ は μ^2 で変わらない. 例えば $u = s_1 s_4$ とおくと,
§1よりu ∈ Q (ζ) だから $u = a + b\zeta + c\zeta^2 + d\zeta^3$ (a, b, c, d ∈ Q) とおける.

ここで「μ^2 によって u は変わらない」ので $a+b\zeta+c\zeta^2+d\zeta^3=$
$a+b\zeta^4+c\zeta^3+d\zeta^2\left(=(a-b)-b\zeta+(-b+d)\zeta^2+(-b+c)\zeta^3\right)$
$1, \zeta, \zeta^2, \zeta^3$ はQ上一次独立だから, $b=0$,
$c=d$. 故に $u=a+c\left(\zeta^2+\zeta^3\right)\in Q\left(\sqrt{5}\right)$. 他も同様.

[補足2] $s_1{}^5$, $s_2{}^5$, $s_3{}^5$, $s_4{}^5$ は 全て実数, または s_1 と s_4, s_2 とs_3 が複素共役

まず, $s_1{}^5=0$ とする.これを μによって移すと,
$s_2{}^5=s_4{}^5=s_3{}^5=0$ となり矛盾. よって $s_1{}^5$, $s_2{}^5$, $s_3{}^5$, $s_4{}^5$ は, どれも 0 ではない.
[補足1] より, $\arg(s_1)=\theta_1$, $\arg(s_2)=\theta_2$ とおくと,
s1 s4, s2 s3 が実数となるから, $\arg(s_3)=-\theta_2$, $\arg(s_4)=-\theta_1$
このとき, $\arg\left(s_1 s_2{}^2\right)=\theta_1+2\theta_2$, $\arg\left(s_3{}^2 s_4\right)=-(\theta_1+2\theta_2)$,
$\arg\left(s_1{}^2 s_3\right)=2\theta_1-\theta_2$, $\arg\left(s_2 s_4{}^2\right)=-(2\theta_1-\theta_2)$
故に $(s_1 s_2{}^2+s_3{}^2 s_4)$ と $(s_1{}^2 s_3+s_2 s_4{}^2)$ が実数より,

「$|s_1 s_2{}^2|=|s_3{}^2 s_4|$ または ($s_1 s_2{}^2$ と $s_3{}^2 s_4$ が実数)」
かつ「$|s_1{}^2 s_3|=|s_2 s_4{}^2|$ または ($s_1{}^2 s_3$ と $s_2 s_4{}^2$ が実数)」
ここで $|s_1 s_2{}^2|=|s_3{}^2 s_4|$ と $|s_2 s_4{}^2|=|s_1{}^2 s_3|$ は μ によって互いに移るので,
片方が成り立てば, もう一方も成り立ち, これから
$|s_2|=|s_3|$ かつ $|s_1|=|s_4|$ となる. このとき s_1 と s_4, s_2 とs_3 が複素共役

$s_1 s_2{}^2$, $s_3{}^2 s_4$, $s_1{}^2 s_3$, $s_2 s_4{}^2$ が実数のときは,
$$\theta_1+2\theta_2\equiv 0 \text{ かつ } 2\theta_1-\theta_2\equiv 0 \to 5\theta_1\equiv 5\theta_2\equiv 0 \pmod{\pi}$$
従って $s_1{}^5$, $s_2{}^5$, $s_3{}^5$, $s_4{}^5$ は 実数となる.

[補足3] $s_1{}^5$, $s_2{}^5$, $s_3{}^5$, $s_4{}^5$ が全て実数のとき, s_1, s_2, s_3, s_4 を実数としても一般性を失わない

$\left(s_1{}^5, s_2{}^5, s_3{}^5, s_4{}^5\right)$ の値の組を一つ固定する. この時 (s_1, s_2, s_3, s_4) の選び方は 5^4 通り.
これらを集めた集合$S=\left\{\left(\zeta^k s_1, \zeta^l s_2, \zeta^m s_3, \zeta^n s_4\right)(k, l, m, n=0, 1, 2, 3, 4)\right\}$
の中の置換μ を次のように定める.
$$\mu: s_1\to\zeta s_1,\ s_2\to\zeta^2 s_2,\ s_3\to\zeta^3 s_3,\ s_4\to\zeta^4 s_4$$

μによって x_i は「$x_1\to x_5\to x_4\to x_3\to x_2\to x_1$」と動く. これは$\tau^3$ によるものと同じ動き. (*)

一方μによって
$(s_1, s_2, s_3, s_4)\to\left(s_1\zeta, s_2\zeta^2, s_3\zeta^3, s_4\zeta^4\right)\to\left(s_1\zeta^2, s_2\zeta^4, s_3\zeta^6, s_4\zeta^8\right)\to$
$\left(s_1\zeta^3, s_2\zeta^6, s_3\zeta^9, s_4\zeta^{12}\right)\to\left(s_1\zeta^4, s_2\zeta^8, s_3\zeta^{12}, s_4\zeta^{16}\right)\to(s_1, s_2, s_3, s_4)$
と動く. 即ち $m=0, 1, 2, 3, 4$ に対し,
μ^m は (s_1, s_2, s_3, s_4) を $\left(s_1\zeta^m, s_2\zeta^{2m}, s_3\zeta^{3m}, s_4\zeta^{4m}\right)$ に移す.

ここで $\arg\zeta = 2\pi/5$ として良い．$s_1{}^5$ は実数だから $\arg s_1$ は $\pi/5$ の整数倍．
故に $\{s_1, s_1\zeta, s_1\zeta^2, s_1\zeta^3, s_1\zeta^4\}$ の中に実数となるものが一つだけ存在する．
それを $s_1\zeta^m$ とすると，$(s_1\zeta^m)(s_4\zeta^{4m}) = s_1 s_4$ が実数より，
$s_4\zeta^{4m}$ も実数．次に $(s_2\zeta^{2m})(s_4\zeta^{4m})^2 = s_2 s_4{}^2$ が実数より $s_2\zeta^{2m}$ も実数．
最後に $(s_2\zeta^{2m})(s_3\zeta^{3m}) = s_2 s_3$ が実数より $s_2\zeta^{2m}$ も実数．
故に $(s_1\zeta^m, s_2\zeta^{2m}, s_3\zeta^{3m}, s_4\zeta^{4m})$ がすべて実数となる m が存在する．

$(s_1\zeta^m, s_2\zeta^{2m}, s_3\zeta^{3m}, s_4\zeta^{4m})$ を改めて (s_1, s_2, s_3, s_4) と置き直すことができる．
なぜなら，$(*)$ より，これは x_1, x_2, x_3, x_4, x_5 の名前を付け替えることに対応するから．

[補足4] f（x）が重解を持たない時，実数解は1個か5個

（ア）$s_1{}^5$，$s_2{}^5$，$s_3{}^5$，$s_4{}^5$ が全て実数のとき．
[補足3] より，s_1, s_2, s_3, s_4 が実数としても一般性を失わない．このとき x_1 は実数．
$\overline{x_2} = \overline{\zeta^4 s_1 + \zeta^3 s_2 + \zeta^2 s_3 + \zeta s_4} = \zeta s_1 + \zeta^2 s_2 + \zeta^3 s_3 + \zeta^4 s_4 = x_5$，同様に $\overline{x_3} = x_4$
故に，実数解が1つと複素共役な虚数解が2組ある．

（イ）s_1 と s_4，s_2 と s_3 が複素共役のとき
Galois群が C_5 のときと同様にして，x_1, x_2, x_3, x_4, x_5 は全て実数．

【参考文献&サイト】次のサイトや文献を参考にさせて頂きました．ありがとうございます．

1. 可解な5次方程式について（大迎　規宏氏）□
2. 方程式のガロア群（松田修氏）□
3. ガロア理論を使って方程式を解いたことがありますか（scruta氏）□
4. ペンギンは空を飛ぶ–　5次方程式の解を巡る旅　（peng225氏）□
5. Period–Mathmatics　可解な5次方程式の代数的解法 □
（以下は書籍）□
6. ガロアの群論（中村亨氏）□
7. ガロワ理論最短コース（梶原　健氏）□
8. ガロア理論の頂きを踏む（石井俊全氏）□
9. ガロア理論「超」入門（小林吹代氏）□

[1] がなんといっても一番詳しいですが難しいです．まだ読了していません．読了したら [1]の方法で解いてみたいと思います．しかし[1] が難しいのは一般論 （例えば Galois群の判別式） をしっかりやっているからで，「有理係数の方程式を解くだけ」なら *Mathematica* を使えば, 結構解けるものだなと感じました．数学的な知識は[6], [7], [8], [9] の入門レベルしか使っていません．また，今回の f (x) の式は[1] から拝借しました．しかし途中の過程はかなり異なっていると思います． それは[1]が理論的な考察を中心としているのに対し，私は方程式を解くことを中心としたからです． 最後に，私は専門家ではないので，色々間違いしているかもしれません．その際はお知らせ下さると有り難いです．なおメールアドレスは　ロボット対策のため 画像となっています．クリックしても何も起こりません.（ mail@mixedmoss.com ）